Collisions with four bullets This is a follow-up question of Colliding Bullets. I'm interested in a rigorous calculation of a specific aspect of the referred question.
We consider four bullets. Once per second a bullet is fired with uniformly random speed in $[0,1]$. If two bullets collide they both disappear.
Let's denote the bullets with $1,2,3$ and $4$ in order of their firing. So, bullet $1$ is fired one second before bullet $2$, etc. Let's denote the speed of bullet $j$ with $v_j, 1\leq j \leq 4$.
We focus on two arrangements:

With probability $\frac{1}{4!}=\frac{1}{24}$ we have
  \begin{align*}
1>v_3>v_4>v_2>v_1>0
\end{align*}
Here are two different scenarios to consider with two different outcomes. One scenario is the collision of $2$ with $1$ leaving two bullets $3$ and $4$ which won't collide, since $3$ is faster than $4$.
  \begin{align*}
3\ 4\ 2\ 1 \rightarrow 3\ 4
\end{align*}
  The other scenario is a collision of $3$ with $2$ and since $4$ is faster than $1$, they will also collide. We obtain
  \begin{align*}
3\ 4\ 2\ 1 \rightarrow 4\ 1\rightarrow \emptyset
\end{align*}
  So, we have two different scenarios, one with two bullets as outcome, while all bullets vanish in the other scenario.

A simulation of these scenarios results in a proportion of roughly
\begin{align*}
28:13
\end{align*}
in favor of a collision of bullet $2$ with bullet $1$, so that bullets $4$ and $3$ will escape to infinity.

A simulation of a similar arrangement with
  \begin{align*}
1>v_3>v_2>v_4>v_1>0
\end{align*}
  results in a proportion of roughly
  \begin{align*}
40:2
\end{align*}
  in favor of a collision of bullet $2$ with bullet $1$.

Question: I'm interested in a rigorous calculation of these proportions. So, an approach with probabilistic or combinatorial methods is highly appreciated. In case you know some numerical solutions, you are heartily invited to add it as comment. 
 A: For small enough versions of this problem, like the 4-bullet case that you're asking for here, you can get exact probabilities by writing down the equations on bullet speeds that give the outcome you're interested in, and then using Mathematica to integrate Boole[those equations] over $[0,1]^4$.
To distinguish combinatorial types of outcomes, the useful equation is that if the bullets fired at times $i$ and $j$ with $i<j$ have velocities $v_i<v_j$, then they will meet at time $\large\frac{j v_j - i v_i}{v_j-v_i}$, as long as they both survive that long.  That equation is not linear, which is why the probabilities are sadly not just volumes of polyhedra.
For the specific example you asked about:
In[1]:= mt[i_,j_] := If[(-i + j)*(-v[i] + v[j]) <= 0,
                        Infinity,
                        (j*v[j] - i*v[i])/(v[j] - v[i])];

In[2]:= Simplify[Integrate[Boole[mt[2,1]<mt[2,3]],
                           {v[1],0,1},{v[2],v[1],1},{v[4],v[2],1},{v[3],v[4],1}]]
Out[2]= (3 - 4*Log[2])/8

In[3]:= Simplify[Integrate[Boole[mt[2,1]>mt[2,3]],
                           {v[1],0,1},{v[2],v[1],1},{v[4],v[2],1},{v[3],v[4],1}]]
Out[3]= (-2 + Log[8])/6

In[4]:= Simplify[%2 + %3]
Out[4]= 1/24

In[5]:= N[%2/%3]
Out[5]= 2.14697

For comparison, your empirical result $28$:$13$ is $2.15385$.
